
Einführung

Bei den parametrisierten Objekten handelt es sich einfach um ein Konstrukt, das es Ihnen ermöglicht, bei einem Methodenaufruf mit einem Objekt mehrere Parameter zu übergeben. Auf diese Weise wird die Methode vereinfacht, da die Anzahl der an die Methode übergebenen Parameter von einer Liste auf einen reduziert wird. Warum sollten Sie dies tun? Sie sehen gleich mehrere Beispiele für den gleichen Methodenaufruf. Das erste Beispiel benutzt die alte Methode einer Liste der Parameter. Im zweiten Beispiel wird ein Parameterobjekt benutzt, um die gleiche Aufgabe zu erfüllen.

Listing 1: Codestück mit einer normalen Parameterliste

goApp.SomeMethod( <cUserName> , ;
   <cUserPassword> , ;
   <dLoginDate> , ;
   <cLoginTime> , ;
   <lAdmin )

Listing 2: Methodenaufruf mit einer normalen Parameterliste

goApp.SomeMethod( "CTB","CTB",DATE(),TIME(),.T.)

Listing 3: Beispiel des gleichen Methodenaufrufs unter Verwendung eines Parameterobjekts:

loParameter = CREATEOBJ("prmParameter")

WITH loParameter

   .cUserName = "CTB"

   .cUserPassword = "CTB"

   .dLoginDate = DATE()

   .cLoginTime = TIME()

   .lAdmin = .T.

ENDWITH

 

goApp.SomeMethod( loParameter )

Nun sollte es klar sein, dass Sie durch die Benutzung eines Parameterobjekts die folgenden Vorteile erlangen: Zunächst einmal wird der Methodenaufruf einfacher, da in ihm nur noch ein einzelner Parameter erscheint. Dadurch erreichen Sie zwei Verbesserungen: 1. die Entwickler müssen die Reihenfolge der Parameter nicht mehr im Kopf haben (also eine einfachere Handhabung) und 2. ist es möglich der Methode zusätzliche Parameter zu übergeben ohne die Schnittstelle der Methode ändern zu müssen (einfachere Verwaltung). Außerdem kommt der Code dem Ideal der Selbstdokumentierung erheblich näher. Woher soll im zweiten Beispiel ein Programmierer zu einem späteren Zeitpunkt wissen, dass es sich bei dem ersten CTB um den Benutzernamen und beim zweiten um das Passwort handelt? In Listing 3 wird klar, für welche Aufgabe welcher Parameter steht. Beachten Sie auch, dass in diesem Beispiel die Reihenfolge der Parameter keine Rolle mehr spielt. Der folgende Code arbeitet genau so gut wie der in Listing 3:

loParameter = CREATEOBJ("prmParameter")

WITH loParameter

   .cUserPassword = "CTB"

   .dLoginDate = DATE()

   .lAdmin = .T.

   .cUserName = "CTB"

   .cLoginTime = TIME()

ENDWITH

Erstellen eines Parameterobjekts als Beispiel

Schritt 1: Erstellen der Klassendefinition des Parameterobjekts

Alle Parameterobjekte des COMCodebook sind von den in IPARAMETEROBJECTS.PRG enthaltenen Superklassen abgeleitet. Dieses Programm finden Sie im Unterverzeichnis \CODEBOOK\COMMON\ILIBS. Es enthält die Klasse lAbstractParameter. Von dieser Klasse sollten Sie alle Parameterobjekte für Ihre Anwendung ableiten, die Sie in einem Layerprogramm abspeichern sollten. Dafür erstellen Sie ein Programm mit dem Namen APARAMETEROBJECTS.PRG, das Sie im Unterverzeichnis LIBS Ihrer Anwendung speichern. Als nächstes erstellen Sie darin die folgende Klassendefinition:

DEFINE CLASS prm_TestParameter AS IAbstractParameter
ENDDEFINE

Schritt 2: Dem Parameterobjekt die Parameter hinzufügen

Sie fügen dem Parameterobjekt Parameter hinzu, indem Sie die Klassendefinition um benutzerdefinierte Parameter erweitern. In unserem Fall benötigen wir fünf Parameter, darum erstellen wir auch fünf benutzerdefinierte Eigenschaften, um die Werte zu speichern. Dafür ändern Sie die Klassendefinition wie folgt:

DEFINE CLASS prm_TestParameter AS IAbstractParameter
cUserPassword = ""
cUserName = ""
dLoginDate = DATE()
cLoginTime = TIME()
lAdmin = .F.

ENDDEFINE

Schritt 3: Dem Parameterobjekt die Parameter bekannt machen

Visual FoxPro enthält keine Funktionalität, die es einem Objekt ermöglicht, Informationen über die Natur, den Typ oder die Anzahl seiner benutzerdefinierten Eigenschaften zu „kennen“. Anders gesagt: Es gibt keine native Möglichkeit einem Objekt die folgenden Fragen zu stellen:

1. Über wie viele benutzerdefinierte Eigenschaften verfügst Du?

2. Wie heißen diese Eigenschaften?

3. Welchen Typ haben sie?

Das Parameterobjekt benutzt ein benutzerdefiniertes Array zusammen mit einigen benutzerdefinierten Methoden um diese Informationen zu speichern und die Fragen beantworten zu können.

Der letzte Schritt bei der Erstellung eines Parameterobjekts ist das Überschreiben der Methode DefineParameterArray() mit der Methode AddParameter(), um dem Parameterobjekt die Informationen über die fünf Parameter „beizubringen“, auf die es reagieren soll. Der letzte Schritt bei der Erstellung des Parameterobjekts ist die folgende Änderung der Klassendefinition, um die Methode goApp.SomeMethod() zu aktivieren.

DEFINE CLASS prm_TestParameter AS IAbstractParameter
cUserPassword = ""
cUserName = ""
dLoginDate = DATE()
cLoginTime = TIME()
lAdmin = .F.

PROCEDURE DefineParameterArray
   DODEFAULT()
   THIS.AddParameter( "cUserName" , "C" )
   THIS.AddParameter( "cUserPassword" , "C" )
   THIS.AddParameter( "dLoginDate" , "D" )
   THIS.AddParameter( "cLoginTime" , "C" )
   THIS.AddParameter( "lAdmin" , "L" )
ENDPROC

ENDDEFINE

Testen des neuen Parameterobjekts

Um das neue Parameterobjekt zu testen müssen Sie zunächst die Bibliotheken einstellen, die für die Unterstützung des Objekts benötigt werden. Dies sind im Einzelnen:

SET PROC TO ..\..\libs\csession.prg

SET PROC TO ..\..\ilibs\isession.prg ADDITIVE

SET PROC TO ..\..\libs\cparameterobjects.prg ADDITIVE

SET PROC TO ..\..\ilibs\iparameterobjects.prg ADDITIVE

SET PROC TO ..\libs\aparameterobjects.prg ADDITIVE

Als nächstes erstellen Sie ein Objekt, genau wie Sie es mit jedem anderen Objekt machen würden.

loParameter = CREATEOBJ('prm_TestParameter')

Zuletzt bestücken Sie das Parameterobjekt mit anderen als den Vorgabewerten.

loParameter.cUserName = "CTB"

loParameter.cPassword = "CTB"

loParameter.lAdmin = .T.

Ein kurzer Blick auf das Locals-Fenster zeigt die Natur des neu erstellten Parameterobjekts an.

